

## SUAVE: Painless Extension for an Object-Oriented VHDL

Peter J. Ashenden  
Dept. Computer Science  
The University of Adelaide, SA 5005  
Australia  
petera@cs.adelaide.edu.au

Philip A. Wilsey and Dale E. Martin  
Dept. ECECS, PO Box 210030  
University of Cincinnati  
Cincinnati, OH 45221- 0030, USA  
phil.wilsey@uc.edu, dmartin@ececs.uc.edu

### Abstract

*The SUAVE project aims to introduce object-oriented extensions to data modeling into VHDL in a way that does not disturb the existing language or its use. Designers regularly define abstract data types by using aspects of VHDL's type system, subprograms, and packages. The SUAVE approach builds on these basic mechanisms by strengthening the facilities for encapsulation and adding an inheritance mechanism. In addition to supporting object-orientation, these extended mechanisms improve the expressiveness of VHDL across the modeling spectrum, from high-level to gate-level. By choosing an incremental and evolutionary approach to extensions, SUAVE avoids major additions to the language that would complicate choice of mechanisms for expressing a design. This paper outlines the SUAVE extensions and illustrates their use through some examples. The mechanisms and examples are readily understood as incremental extensions to current modeling practices, hence "painless extension."*

### 1. Introduction

VHDL is widely used by designers of digital systems for specification, simulation and synthesis. Increasingly, designers are using VHDL at high levels of abstraction as part of the system-level design process. At this level of abstraction, the aggregate behavior of a system is described in a style that is similar to that of software. Data is modeled in abstract form, rather than using any particular binary representation, and functionality is expressed in terms of interacting processes that perform algorithms of varying complexity. A subsequent partitioning step in the design process may determine which aspects of the modeled be-

havior are to be implemented as hardware subsystems, and which are to be implemented as software.

Experience in the software engineering community has lead to adoption of object-oriented design and programming techniques for managing complexity through abstract data types (ADTs) and re-use [8]. Features included in programming languages to support these techniques are abstraction and encapsulation mechanisms, inheritance, and genericity. The term "object-based" is widely used to refer to a language that included abstraction and encapsulation mechanisms [21]. The term "object-oriented" is used to refer to a language that additionally includes inheritance.

While VHDL can be used for modeling at the system level, it has some deficiencies that make the task more difficult than it would otherwise be. These difficulties center around language features (or lack of some features) for supporting complexity management. VHDL is currently somewhat less than object-based, as its encapsulation mechanism are weak. It is certainly not object-oriented, as it does not include any form of inheritance. While it does include a mechanism for genericity, that mechanism is severely limited, allowing only parameterization of units by constant values. We have discussed these issues in a previous paper [2].

SUAVE aims to improve support for high-level modeling in VHDL by extending the language with features for object-orientation and genericity in a way that does not disturb the existing language or its use. As well as adding specific language features, some existing features are generalized, the facilities for encapsulation are strengthened, and an inheritance mechanism is added. Private types and private parts in packages support improved encapsulation. Type derivation, record type extension, and class-wide types with dynamic dispatching support inheritance. We have previously argued [3] that, in addition to

---

\* This work was partially supported by Wright Laboratory under USAF contract F33615- 95- C- 1638.

supporting object-orientation, these extensions improve the expressiveness of VHDL across the modeling spectrum from high-level to gate-level.

By choosing an incremental and evolutionary approach to extensions, SUAVE avoids major additions to the language that would complicate choice of mechanisms for expressing a design. Furthermore, the implementation burden is not large, and there is no performance penalty in simulation or synthesis if the mechanism are not used. The SUAVE approach is similar to that proposed by Mills [15] and by Schumacher and Nebel [18]. It is contrasted with others that have been proposed [7, 10, 17, 20, 22], that add new, separate mechanisms for combining abstraction, encapsulation, and inheritance for object-orientation. Such mechanisms replicate aspects of the existing features of VHDL, making design choices for expressing a model more complex.

This paper outlines the SUAVE extensions for object-orientation and illustrates their use through some examples. (The SUAVE extensions for genericity are described in a companion paper [4]. More complete presentation of the extensions can be found in the SUAVE report [5].) Most of the features added to VHDL are adapted from features in Ada-95 [14], and are included largely for the same reasons that they are included in Ada-95 [6]. Section 2 of this paper outlines the design principles and objectives that were followed in deciding how to extend VHDL. Subsequent sections describe the extensions in detail and illustrate them with examples. Section 3 describes the extensions to the type system of VHDL to support type derivation, extension and class-wide programming. Section 4 describes the extensions that improve the encapsulation features of VHDL. In combination, the extensions in these two sections turn VHDL into an object-oriented language. We conclude in Section 5 by summarizing how the mechanisms and examples are readily understood as incremental extensions to current modeling practices, hence “painless extension.”

## 2. SUAVE Design Objectives

A previous paper [3] reviews the issues to be addressed in extending VHDL for high-level modeling and discusses principles that should govern the design of language extensions. As a result of that analysis, a number of design objectives were formulated for SUAVE:

**Sto** improve support for high-level behavioural modeling by improving encapsulation and information hiding capabilities and providing for hierarchies of abstraction,

**Sto** improve support for re-use and incremental development by allowing further delaying of bindings through type-genericity and dynamic polymorphism,

**Sto** preserve capabilities for synthesis and other forms of design analysis,

**Sto** support hardware/software co-design through improved integration with programming languages (e.g., Ada),

**Sto** support refinement of models through elaboration of components rather than through repartitioning, and

**Sto** preserve correctness of existing models within the extended language.

Since SUAVE is an extension of the existing VHDL language, it is important that the extensions integrate well with all aspects of the existing language. In designing the SUAVE extensions, the design principals followed during the restandardization of VHDL that lead to the current language [13] were adopted in addition to those listed above. The goal was to preserve what Brooks refers to as the “conceptual integrity” of the language [9].

## 3. Extensions to the Type System

Object-oriented languages support re-use and incremental development through the mechanism of inheritance. SUAVE extends the type system of VHDL by adopting the object-oriented features of Ada-95, including inheritance through type derivation and tagged types with extension on derivation.

### 3.1 Derived Types and Inheritance

For a type defined in a package, the operations (procedures and functions) defined in the package are called the *primitive operations* of the type. A new type can be defined as being *derived* from a parent type. In that case, the derived type inherits the set of values and the primitive operations of the parent type. An inherited operation can be overridden by defining a new operation with the same name but with operands of the derived type. Furthermore, additional primitive operations can be defined for the derived type. SUAVE adopts the Ada notation for defining a derived type, for example:

```
type event_count is new natural;
```

The derived type is distinct from, but related to, the parent type. Use of derived types helps avoid inadvertent mixing of conceptually different values, and thus improves the expressiveness of the language.

### 3.2 Tagged Types and Type Extension

As in Ada-95, a record type in SUAVE may include the reserved word **tagged** in its definition. Such a type is called a *tagged type*. An object of a tagged type includes a run-

time tag that identifies the specific type used to create the object. The tag is used for dynamic dispatching, which is described below. A tagged record type may be *extended* by deriving a new type with a *record extension* containing additional record elements. This is the origin of the term “programming by extension,” sometimes used to describe the Ada-95 approach. The derived type is also a tagged type that can be further extended. Since all elements in the parent type are also in the derived type, inherited operations of the parent type can be applied to objects of the extended type. However, any overriding or newly defined operations for the extended type can only be applied to the extended type (or its derivatives), since they may refer to the elements in the extension.

As an example, consider a type and operations representing an instruction set for a RISC CPU. All instructions have an opcode. ALU instructions additionally have fields for the source and destination register numbers. Thus an ALU instruction can be considered as an extension of a base instruction with just an opcode. This can be expressed in SUAVE by defining the following in a package:

```

type instruction is
  tagged record
    opcode : opcode_type;
  end record instruction;

function privileged ( instr : instruction;
                    mode : protection_mode )
  return boolean;

procedure disassemble ( instr : instruction;
                      file output : text );

type ALU_instruction is
  new instruction with record
    destination,
    source_1, source_2 : register_number;
  end record ALU_instruction;

procedure disassemble ( instr : ALU_instruction;
                      file output : text );

```

The subprograms `privileged` and `disassemble` are primitive operations of instruction and are inherited by derived types. The type `ALU_instruction` is derived from `instruction` and has four elements: the opcode element inherited from `instruction`, and the three register number elements defined in the extension. A version of the function `privileged` is inherited from `instruction` with the `instr` parameter being of type `ALU_instruction`. The `disassemble` instruction defined for `ALU_instruction` overrides that inherited from `instruction`.

### 3.3 Abstract Types and Subprograms

An *abstract type* is a tagged type that is intended for use solely as the parent of some other derived type. Objects

may not be declared to be of an abstract type. An *abstract subprogram* is one that has no body (and requires none), because it is intended to be overridden when inherited by a derived type. Abstract types and subprograms allow definition of types that include common properties and operations, but which must be refined by derivation of types that represent concrete objects.

As an illustration, consider refinement of the instruction type to represent memory reference instructions using displacement addressing mode. Such instructions include a base register number and an offset. The type for these instructions is declared abstract, since it is intended to be the parent type for load and store instruction types. More precisely,

```

type memory_instruction is
  abstract new instruction with record
    base : register_number;
    offset : integer;
  end record memory_instruction;

function effective_address_of
  ( instr : memory_instruction );

procedure perform_memory_transfer
  ( instr : memory_instruction ) is abstract;

```

The function `effective_address_of` is not abstract, since it can calculate the result using the data in a `memory_instruction` record. The function can be inherited “as is” by derived types. The procedure `perform_memory_transfer`, on the other hand, is declared abstract since the direction of transfer depends on whether a memory instruction is a load or a store. The derived types must provide overriding non-abstract implementations of this procedure. Examples are derived types for load and store instructions, as follows:

```

type load_instruction is
  new memory_instruction with record
    destination : reg_number;
  end record load_instruction;

procedure perform_memory_transfer
  ( instr : load_instruction );

procedure disassemble ( instr : load_instruction;
                      file output : text );

type store_instruction is
  new memory_instruction with record
    source : reg_number;
  end record store_instruction;

procedure perform_memory_transfer
  ( instr : store_instruction );

procedure disassemble ( instr : store_instruction;
                      file output : text );

```

Objects cannot be declared to be of type `memory_instruction`, but they can be declared to be of type `load_instruction` or `store_instruction`.

### 3.4 Class-Wide Types and Operations

One of the most important aspects of object-oriented programming is the use of *classes*. SUAVE adopts the Ada-95 mechanism of *class-wide types* to deal with classes. This contrasts with languages such as Simula [11], C++ [19] and Java [12] that introduce a special construct for classes. (See our paper that compares the two approaches [1].)

Class-wide types are denoted using the 'Class attribute. For a tagged type T, the *class-wide type* denoted T'Class is the union of T and all types derived directly or indirectly from T. The type T is called the *root* of the class-wide type. For example, the class-wide type instruction'class denotes the hierarchy of types rooted at instruction, and including ALU\_instruction, memory\_instruction, load\_instruction and store\_instruction.

An object of a class-wide type can have a value of any specific type in T'Class. Such an object is called *polymorphic*, meaning that it can take on values of different types during its lifetime. SUAVE allows constants, dynamically allocated variables and signals to be of a class-wide type. When an operation is applied to an object of a class-wide type, the tag of the value is used to determine the specific type, and thus to determine which primitive operation to invoke. This is called *dynamic dispatching*, or *late binding*, and is an essential aspect of object-oriented languages. As an example, consider the following signal declaration and application of an operation:

```
signal fetched_instruction : instruction'class;
disassemble ( fetched_instruction );
```

If the value of the signal is of type instruction, the version of disassemble for that type is invoked. However, if the value of the signal is of one of type load\_instruction, the overriding version defined for load\_instruction values is invoked. The choice is made dynamically at the time of the call.

While there are no primitive operations of a class-wide type, a subprogram may have a parameter of a class-wide type. Such a subprogram is called a *class-wide operation*. For example:

```
procedure execute ( instr : instruction'class );
```

Since the parameter is polymorphic, dynamic dispatching may be required for operations on the parameter within the subprogram.

As a final example in this section, consider an instruction register that can jam a TRAP instruction in place of the store instruction. First, two constants are declared for the TRAP instruction and an undefined instruction:

```
constant halt_instruction : instruction
:= instruction'(opcode => op_halt);
constant undef_instruction : instruction
:= instruction'(opcode => op_undef);
```

Next, the entity is declared:

```
entity instruction_reg is
port ( load_enable : in bit;
      jam_halt : in bit;
      instr_in : in instruction'class;
      instr_out : out instruction'class );
end entity instruction_reg;
```

The ports instr\_in and instr\_out are signals of a class-wide type and so may take on values of any of the types in the instruction hierarchy. A behavioral architecture body for the register is:

```
architecture behavioral of instruction_reg is
begin
store : process ( load_enable, jam_halt,
                 instr_in ) is
type instruction_ptr is
access instruction'class;
variable stored_instruction : instruction_ptr
:= new undef_instruction;
begin
if jam_halt = '1' then
deallocate ( stored_instruction );
stored_instruction := new halt_instruction;
elsif load_enable = '1' then
deallocate ( stored_instruction );
stored_instruction := new instr_in;
end if;
instr_out <= stored_instr.all;
end process store;
end architecture behavioral;
```

The process implements the register storage using the local variable stored\_instruction. Since a variable cannot be of a class-wide type, stored\_instruction is defined as an access value, pointing to a dynamically allocated object of type instruction'class. It is initialized to the undefined instruction. When a HALT instruction is to be jammed, a new instruction object initialized to the halt instruction value is allocated. Similarly, when an input instruction is to be stored, a new instruction object of the corresponding specific type is allocated and initialized to the input instruction. The designated instruction object is assigned as the output of the register.

## 4. Extensions for Encapsulation

A data type in VHDL is characterized by a set of values, specified by a type definition, and a set of operations. An *abstract data type* (ADT) is one in which the concrete details of the type definition are hidden from the user of the

ADT. The user may only use the operations of the ADT to manipulate values. ADTs are important tools for managing complexity in a large design.

VHDL currently includes the *package* feature, which can be used to define an ADT. The concrete type and associated operations are declared in the package declaration, and the implementations of the operations are declared in the package body. While this approach allows the implementation details of the operations to be hidden from the ADT user, it exposes the details of the concrete type. A user may inadvertently (or deliberately) modify values of the concrete type directly, rather than by using the provided operations. This can potentially place the ADT value in an inconsistent state. It also reduces the maintainability of the design.

SUAVE extends the type system and package feature of VHDL to provide secure encapsulation of information in an ADT. It adopts the mechanisms of private types and private parts in packages from Ada-95. This meets one of the design objectives for SUAVE: to improve encapsulation and information hiding.

As a first step, the use of packages is generalized by allowing them to be declared as part of most declarative regions in a model, not just as library units. SUAVE allows a package declaration and body to be declared in an entity declaration, an architecture body, a block statement, a generate statement, a process statement, and a subprogram body. Thus, the concept of a package is changed from that of a “heavy-weight” library-level unit to that of a “light-weight” declarative item. This is important, since packages are used to declare types and operations defining classes, as well as instances of generic packages (see Ashenden *et al* [4]).

#### 4.1 Private Parts and Private Types

The second extension of the package feature is to allow a package declaration to be divided into a *visible part* and a *private part*, as follows:

```
package name is
  ...    -- visible part
private
  ...    -- private part
end package name;
```

Items declared in the visible part are exported and may be referred to by users of the package. Items declared in the private part, on the other hand, are not visible outside the package. When using a package to define an ADT, the type is declared as a *private type* in the visible part of the package, along with the specifications of the primitive operations of the type. A private type declaration only provides

the name of the type. The concrete details of the type are declared separately in the private part of the package.

As an example, the following package defines an ADT for complex numbers:

```
package complex_numbers is
  type complex is private;
  constant i : complex;
  function cartesian_complex ( re, im : real )
    return complex;
  function re ( C : complex ) return real;
  function im ( C : complex ) return real;
  function polar_complex ( r, theta : real )
    return complex;
  function "abs" ( C : complex ) return real;
  function arg ( C : complex ) return real;

  function "+" ( L, R : complex ) return complex;
  function "-" ( L, R : complex ) return complex;
  function "*" ( L, R : complex ) return complex;
  function "/" ( L, R : complex ) return complex;
private
  type complex is
    record
      r, theta : real;
    end record complex;
end package complex_numbers;
```

A user of this package can declare objects of type complex and invoke operations on complex numbers, for example:

```
signal x, y, z : complex
  := cartesian_complex(0.0, 0.0);
...
z <= x * y after 20 ns;
```

However, the fact that complex numbers are represented in polar form is hidden. Indeed, the representation may be changed without requiring changes to the user’s code.

#### 4.2 Private Extensions

SUAVE adopts the Ada-95 mechanisms for integrating encapsulation with inheritance. A private type can be declared to be tagged, indicating that it can be used as the parent of a derived type. The concrete details remain hidden in the private part of the package. A tagged private type can also be declared abstract if it should not be directly instantiated. For example, a network packet at the media-access level of a protocol suite might be declared as follows:

```
package MAC_level is
  type MAC_packet is abstract tagged private;
  ...
private
```

```

type MAC_packet is tagged record
    ...
end record MAC_packet;
end package MAC_level;

```

A tagged private type can be extended using type derivation, as described in Section 3. However, for the derived type to take on the form of a secure ADT, it should be declared as a *private extension*. This allows the details of the extension to be encapsulated. For example, the network packet type defined above may be extended with payload information to form a network-level packet:

```

package network_level is
    type network_packet is
        new MAC_packet with private;
    ...
private
    type network_packet is
        new MAC_packet with record
            ...
        end record network_packet;
end package network_level;

```

A user of this package knows that a network-level packet is derived from a MAC-level packet, and thus inherits all of the operation applicable to a MAC-level packet. The concrete details of both types, however, remain hidden.

### 4.3 Contractual Details

In adopting the Ada-95 features for private types into VHDL, some minor changes were required to take account of interactions with VHDL-specific features. In particular, VHDL prohibits signals from being of a type that includes access values. The reason for the restriction is that signals are the communication medium between processes, which execute concurrently. If processes were to pass access values between one another, the designated variable would be shared and thus liable to uncontrolled concurrent access. Furthermore, in a parallel implementation of a simulator, different processes may execute in different address spaces or on different processors. An access value created in one process may be meaningless in the addressing context of another.

SUAVE requires that a private type whose concrete implementation includes an access value to indicate the fact in the private type declaration with the keywords **access private**. The same requirement applies to a private extension that includes an access value. Such types cannot be used for signals. Indication of the existence of an access type in the concrete type can be viewed as a form of contract between the type provider and users. Absence of the indication is contract that the concrete type does not in-

clude access values. In the case of a signal of a class-wide type, there may be a derived type in the class that includes an access value. While this cannot be checked during analysis, it can be determined at elaboration time, since the complete hierarchy covered by the class is known at that time.

Another form of contract that can be specified relates to assignment. If a private type includes the keyword **limited**, assignment is not allowed by the user of the type, and the equality operator is not predefined. This feature is adopted from Ada, and is useful for types denoting linked data structures. Assignment normally involves element-wise copying of values, and equality involves element-wise comparison. For linked structures, deep copy and deep comparison may be more appropriate. The type is declared limited in the visible part of the package, and copy and equality operations are provided. The implementations of the operations have full view of the type, and so can implement the required deep copy and comparison.

As an example of the two forms of contractual detail described in this section, consider the following ADT for a set of test vectors:

```

package test_vector_lists is
    type list is limited access private;
    constant empty_list : list;

    procedure copy ( from : in list; to : out list );
    impure function "=" ( L, R : list ) return boolean;
    procedure add ( L : inout list;
                  test : in test_vector );
    ...
private
    type element_node;
    type element_ptr is access element_node;
    type list is new element_ptr;
end package test_vector_lists;

```

The list type is represented by the private type list, whose concrete representation is a singly-linked list of elements. Since the type includes access values, the keyword **access** is included in the private type declaration. Further, since the intended semantics of list assignment is to copy the elements to the target, the private type is made limited. Hence the package provides a copy operation and an equality operation. The body of the package is outlined as follows.

```

package body test_vector_lists is
    type element_node is record
        next_element : element_ptr;
        element : test_vector;
    end record element_node;
    constant empty_list : list
        := list ( element_ptr'(null) );

```

```

    procedure copy ( from : in list;
                   to : out list ) is ...
    ...
end package body test_vector_lists;

```

This illustrates a further extension to VHDL made by SUAVE: constants and constant parameters may include access values. This improves the expressiveness of the language by allowing constants to be of an ADT whose implementation happens to include access values. It also allows operations of such an ADT to be written functions with constant in-mode parameters of the type and a result of the type.

## 5. Conclusion

In this paper we have described the SUAVE extensions to VHDL to improve its support for modeling at all levels of abstraction. We have presented the features that provide object-orientation as a combination of improved abstraction, encapsulation and inheritance mechanisms. We describe the new features for genericity in a companion paper [4]. Most of the features are drawn from Ada-95 and are adapted to integrate with modeling features that are specific to VHDL. Drawing on Ada is appropriate, since VHDL was originally strongly influenced by Ada. In a sense, SUAVE is an evolution of VHDL that parallels the evolution from Ada-83 to Ada-95.

SUAVE improves modeling support by generalizing and extending existing mechanisms, rather than by adding whole new features, hence the suggestion that SUAVE is a "painless extension." In particular, SUAVE avoids replication of the abstraction & encapsulation mechanisms already provided by the package feature. Adding a separate class feature, as proposed in Objective VHDL [17], for example, replicates many aspects of packages and so complicates a designer's choice of expression of design intent.

Space considerations preclude a more detailed definition of the features added in SUAVE. The interested reader can find a more complete description in the SUAVE report [5]. Work is now in progress to implement the extensions within the framework of the SAVANT project [16].

## References

- [1] P. J. Ashenden and P. A. Wilsey, *A Comparison of Alternative Extensions for Data Modeling in VHDL*, Dept. Computer Science, University of Adelaide, Technical Report TR-02/97, <ftp://ftp.cs.adelaide.edu.au/pub/VHDL/TR-data-modeling.ps>, 1997.
- [2] P. J. Ashenden and P. A. Wilsey, "Considerations on Object-Oriented Extensions to VHDL," *Proceedings of VHDL International Users Forum Spring 1997 Conference*, Santa Clara, CA, pp. 109- 118, 1997.
- [3] P. J. Ashenden and P. A. Wilsey, *Principles for Language Extension to VHDL to Support High-Level Modeling*, Dept. Computer Science, University of Adelaide, Technical Report TR-03/97, <ftp://ftp.cs.adelaide.edu.au/pub/VHDL/TR-principles.ps>, 1997.
- [4] P. J. Ashenden and P. A. Wilsey, "Reuse Through Genericity in SUAVE," *Proceedings of VHDL International Users Forum Fall 1997 Conference*, Washington, DC, 1997.
- [5] P. J. Ashenden, P. A. Wilsey, and D. E. Martin, *SUAVE Proposal for Extensions to VHDL for High-Level Modeling*, Dept. Computer Science, University of Adelaide, Technical Report to be published, <ftp://ftp.cs.adelaide.edu.au/pub/VHDL/TR-extension.ps>, 1997.
- [6] J. Barnes, Ed. *Ada 95 Rationale*, vol. 1247. Berlin, Germany: Springer-Verlag, 1997.
- [7] J. Benzakki and B. Djaffri, "Object Oriented Extensions to VHDL: the LaMI Proposal," *Proceedings of Conference on Hardware Description Languages '97*, Toledo, Spain, pp. 334- 347, 1997.
- [8] G. Booch, *Object-Oriented Analysis and Design with Applications*. Redwood City, CA: Benjamin/Cummings, 1994.
- [9] F. P. Brooks, Jr., *The Mythical Man-Month*, Anniversary ed. Reading, MA: Addison-Wesley, 1995.
- [10] D. Cabanis and S. Medhat, "Classification-Oriented for VHDL: A Specification," *Proceedings of VHDL International Users Forum Spring '96 Conference*, Santa Clara, CA, pp. 265- 274, 1996.
- [11] O. J. Dahl and K. Nygaard, "Simula: An Algol Based Simulation Language," *Communications of the ACM*, vol. 9, no. 9, pp. 671- 678, 1966.
- [12] J. Gosling, B. Joy, and G. L. Steele, *The Java Language Specification*. Reading, MA: Addison-Wesley, 1996.
- [13] IEEE, *Standard VHDL Language Reference Manual*. Standard 1076-1993, New York, NY: IEEE, 1993.
- [14] ISO/IEC, *Ada 95 Reference Manual*. International Standard ISO/IEC 8652:1995 (E), Berlin, Germany: Springer-Verlag, 1995.
- [15] M. T. Mills, *Proposed Object Oriented Programming (OOP) Enhancements to the Very High Speed Integrated Circuits (VHSIC) Hardware Description Language (VHDL)*, Wright Laboratory, Dayton, OH, Tech. Report WL-TR-5025, 1993.
- [16] MTL Systems Inc., *Standard Analyzer of VHDL Applications for Next-generation Technology (SAVANT)*. MTL Systems, Inc, <http://www.mtl.com/projects/savant/>, 1996.

- [17] M. Radetzki, W. Putzke, W. Nebel, S. Maginot, J.-M. Bergé, and A.-M. Tagant, "VHDL Language Extensions to Support Abstraction and Re-Use," *Proceedings of Workshop on Libraries, Component Modeling, and Quality Assurance*, Toledo, Spain, 1997.
- [18] G. Schumacher and W. Nebel, "Inheritance Concept for Signals in Object-Oriented Extensions to VHDL," *Proceedings of Euro-DAC '95 with Euro-VHDL '95*, Brighton, UK, pp. 428- 435, 1995.
- [19] B. Stroustrup, *The C++ Programming Language*. Reading, MA: Addison-Wesley, 1986.
- [20] S. Swamy, A. Molin, and B. Covnot, "OO-VHDL: Object-Oriented Extensions to VHDL," *IEEE Computer*, vol. 28, no. 10, pp. 18- 26, 1995.
- [21] P. Wegner, "Dimensions of Object-Based Language Design," *ACM SIGPLAN Notices*, vol. 22, no. 12, *Proceedings of OOPSLA '87*, pp. 168- 182, 1987.
- [22] J. C. Willis, S. A. Bailey, and R. Newschutz, "A Proposal for Minimally Extending VHDL to Achieve Data Encapsulation Late Binding and Multiple Inheritance," *Proceedings of VHDL International Users Forum Fall '94 Conference*, McLean, VA, pp. 5.31- 5.38, 1994.